TP2 : Calcul du $m^{\text{ème}}$ élément, des m premiers éléments d'une suite

(Révisions sur la structure itérative `for`*)*

Pré-requis : l'objectif des premières séances de TP est de faire le point sur des fonctionnalités importantes de **Scilab** qui ont été vues en première année. Je vous invite à consulter les chapitres de cours correspondants sur ma page : support informatique.
On pourra en particulier se reporter au « CH 7 : Les structures itératives ».

▶ Dans le dossier `Info_2a` (resp. `Info_3a`) créé au TP précédent, créer le dossier `TP_2`.

I. Calcul du $10^{\text{ème}}$ élément d'une suite

On considère la suite $(u_n)_{n\in\mathbb{N}^*}$ définie par : $\begin{cases} \forall n \in \mathbb{N}^*,\ u_{n+1} = 2\,u_n + n + 1 \\ u_1 = 1 \end{cases}$

D'aute part, on considère le programme **Scilab** suivant :

```
1  u = 1
2  u = 2 * u + 1 + 1
3  u = 2 * u + 2 + 1
4  u = 2 * u + 3 + 1
5  u = 2 * u + 4 + 1
6  u = 2 * u + 5 + 1
```

▶ Que réalise ce programme ?

- À la fin de l'exécution de la première ligne, la variable `u` contient la valeur de u_1.
- La valeur de u_2 est donnée par la formule : $u_2 = 2\,u_1 + 1 + 1$.
  La deuxième ligne consiste à mettre à jour la variable `u` en lui assignant la valeur calculée par `2 * u + 1 + 1`. Or en début de ligne 1, la variable `u` contient u_1.
  Ainsi, en fin de ligne, une fois l'affectation exécutée, la variable `u` contient la valeur de u_2.
- Ainsi, `u` contient les valeurs successives de la suite (u_n).

▶ Comment obtenir la valeur de u_{10} ? De u_{20} ? De u_{250} ?

- Pour calculer u_{10} et u_{20}, on peut envisager d'ajouter des lignes au programme précédent.
- Cette méthode ne convient évidemment pas pour le calcul de u_{250}. On fait alors appel à une structure itérative. Plus précisément, étant donné que l'on connaît le nombre d'itérations, on utilise une boucle `for`.

```
1  u = 1
2  for i = 1:249
3    u = 2 * u + i + 1
4  end
```

- On remarque au passage que pour obtenir u_{250} on initialise la variable `u` pour lui donner la valeur u_1 puis on effectue 249 (attention à ne pas écrire 250) mises à jour.

II. Calcul du $m^{\text{ème}}$ élément d'une suite

- Dans un onglet **SciNotes**, écrire un programme qui :
  - × demande initialement à l'utilisateur d'entrer au clavier la valeur d'un entier `m`,
  - × initialise une variable `u` à la valeur 1,
  - × met à jour `u` dans une structure itérative de sorte à ce que `u` contienne la valeur du $m^{\text{ème}}$ élément de la suite en fin de boucle.
  - × affiche la valeur de `u`.

  Sauvegarder ce programme sous le nom `emeSuiteU.sce`.

```
m = input("Prière d''entrer un entier m : ")
u = 1
for i = 1:m-1
    u = 2 * u + i + 1
end
disp(u)
```

- Calculer u_{12} et u_{20} à l'aide du programme précédent.

On obtient $u_{12} = 8178$ et $u_{20} = 2097130$.

- Dans un nouvel onglet **SciNotes**, copier-coller le programme précédent.
  Modifier ce programme afin d'obtenir une fonction `emeSuiteU` qui :
  - × prend en paramètre une variable `m`,
  - × calcule en sortie une variable `u` contenant le `m`$^{\text{ème}}$ élément de la suite (u_n).

```
function u = emeSuiteU(m)
    u = 1
    for i = 1:m-1
        u = 2 * u + i + 1
    end
endfunction
```

- Calculer u_7 et u_{15} à l'aide de la fonction précédente.

On obtient $u_7 = 247$ et $u_{15} = 65519$.

- Selon vous, quels sont les avantages de la représentation sous forme de programme avec dialogue utilisateur ? Sous forme de fonction ?

- D'un point de vue utilisateur, la version avec dialogue utilisateur, plus ludique, peut être plus appréciée.
- D'un point de vue algorithmique, il faut privilégier la version sous forme de fonction. L'avantage est que le calcul réalisé par `emeSuiteU` peut facilement être utilisé ailleurs (notamment dans une autre fonction) : il suffit pour ce faire d'écrire l'appel `emeSuiteU(m)` (avec `m` choisi correctement).

III. Calcul des m premiers éléments d'une suite

- Dans un onglet **SciNotes**, écrire un programme qui :
  - × demande initialement à l'utilisateur d'entrer au clavier la valeur d'un entier `m`,
  - × crée un vecteur `U` composé initialement de `m` zéros,
  - × met à jour les cœfficients de `U` dans une structure itérative de sorte à ce que `U` contienne les valeurs des m premiers éléments de la suite (u_n) en fin de boucle.

  Sauvegarder ce programme sous le nom `premSuiteU.sce`.

```
m = input("Prière d''entrer un entier m : ")
U = zeros(1,m)
U(1) = 1
for i = 1:m-1
    U(i+1) = 2 * U(i) + i + 1
end
```

- Calculer les 5 premiers éléments de la suite à l'aide du programme précédent.

On obtient : $[1, 4, 11, 26, 57]$.

- Dans un nouvel onglet **SciNotes**, copier-coller le programme précédent.
  Modifier ce programme afin d'obtenir une fonction `premSuiteU` qui :
  - × prend en paramètre une variable `m`,
  - × calcule en sortie une variable `U` contenant les `m` premiers éléments de la suite (u_n).

```
function U = premSuiteU(m)
    U = zeros(1, m)
    U(1) = 1
    for i = 1:m-1
        U(i+1) = 2 * U(i) + i + 1
    end
endfunction
```

- Effectuer le tracé des 30 premières valeurs de la suite (u_n).
  Les points correspondant devront apparaître sous la forme d'un cercle rouge.
  Quelle conjecture peut-on émettre sur la limite de la suite (u_n) ?

```
N = 1:30
U = premSuiteU(30)
plot(N, U, 'ro')
```

Au vu du graphique obtenu, on peut conjecturer que la suite (u_n) tend vers $+\infty$.
(il est par exemple simple de démontrer que, pour tout $n \in \mathbb{N}$, $u_n \geqslant 2^{n-1}$)

IV. Suites $u_{n+1} = f(u_n)$ aux concours

IV.1. ECRICOME - 2015

L'épreuve **ECRICOME - 2015** commençait par l'étude d'une suite récurrente de type $u_{n+1} = F(u_n)$. La fonction F est définie comme suit :

$$F(x) = \begin{cases} 0 & \text{si } x < 0, \\ 1 - \mathrm{e}^{-x} & \text{si } x \geqslant 0 \end{cases}$$

On considère la suite $(u_n)_{n\geqslant 1}$ définie par $u_1 = 1$ et pour tout $n \in \mathbb{N}^*$ par : $u_{n+1} = F(u_n)$.

▶ *Recopier et compléter le programme* ***Scilab*** *suivant qui permet de représenter les cent premiers termes de la suite* $(u_n)_{n\geqslant 1}$:

```
1 U = zeros(1,100)
2 U(1) = 1
3 for n = 1 : 99
4   U(n+1) = -------------
5 end
6 plot(U,"+")
```

(on demandait de démontrer, dans une question précédente que : $\forall n \geqslant 1,\ u_n > 0$)

```
U(n+1) = 1 - exp(U(n))
```
(lorsqu'il n'y a qu'une ligne à compléter, il n'y a pas lieu de recopier tout le programme)

▶ *Le programme précédent complété permet d'obtenir la représentation graphique suivante.*



Quelle conjecture pouvez-vous émettre sur la monotonie et la limite de la suite $(u_n)_{n\geqslant 1}$ *?*

D'après la représentation graphique, la suite (u_n) semble être convergente, de limite nulle.

IV.2. ESSEC II - 2016

L'épreuve **ESSEC II - 2016** comportait une unique question d'informatique [1] qui consistait au codage d'une suite récurrente définie comme suit :

$$\begin{cases} u_0 = 1 \\ u_n = u_{n-1}\, p_1 + \ldots + u_0\, p_n \end{cases}$$

▶ *En **Scilab**, soit $P = [p_1, p_2, \ldots, p_n]$ le vecteur ligne tel que $P(j) = p_j$ pour j dans $[\![1, n]\!]$. Écrire un programme en **Scilab** qui calcule u_n à partir de P.*

Remarque

Cette question est bien plus compliquée que ce que laisse entrevoir son énoncé. Détaillons pourquoi.

1) En terme de structures de données :

La suite (u_n) est une suite récurrente dont le $n^{\text{ème}}$ terme dépend de `TOUS` les précédents.
Ce type de suite est bien plus délicate à traiter que les suites récurrentes d'ordre 1 (dont la relation de récurrence s'écrit $u_{n+1} = f(u_n)$) ou d'ordre 2 ($u_{n+1} = g(u_n, u_{n-1})$).

Pour calculer le terme d'indice n, il faut avoir accès aux termes d'indice $0, \ldots, n-1$ de la suite.
↪ il faut donc se servir d'un vecteur pour stocker au fur et à mesure ces valeurs.

2) En terme d'indices :

La suite (u_n) est définie à partir du rang 0 alors que la numérotation des cœfficients matriciels commence à 1 en **Scilab**. De plus, la formule de récurrence mélange ordre croissant pour les éléments de (u_n) et décroissant pour ceux de (p_n).

3) En terme de paramètre :

La fonction consiste à calculer le $n^{\text{ème}}$ terme de la suite (u_n) mais n n'est pas annoncé comme paramètre de la fonction. En effet, le seul paramètre annoncé pour la fonction est le vecteur P. C'est la taille de ce vecteur qui nous fournit la valeur de n.

Pour ce type de questions, il est conseillé de commencer par écrire au brouillon les premières étapes de l'algorithme. Autrement dit, d'effectuer le calcul de u_0, u_1, u_2, u_3, afin de comprendre le mécanisme de calcul.

```
1  function res = calcSuiteU(P)
2      [m, n] = size(P) // ou n = length(P)
3      U = zeros(1, n+1) // il y a n+1 entiers entre 0 et n
4      U(1) = 1 // la première case contient la valeur de u0
5      for i = 1:n
6          aux = 0 // variable auxiliaire
7          for j = 1:i
8              aux = aux + U(j) * P((i+1)-j)
9          end
10         U(i+1) = aux
11     end
12     res = U(n+1)
13 endfunction
```

En fin de programme, la variable `res` contient u_n. Comme l'exige l'énoncé, c'est cette valeur qui est renvoyée et pas le vecteur `U` des n premiers éléments de la suite (u_n).

[1] Ce qui représente une nette amélioration par rapport à l'épreuve 2015.

Afin de répondre à cette question, on pouvait aussi observer que :

$$u_{n-1}\,p_1 + \ldots + u_0\,p_n = \begin{pmatrix} u_0 & u_1 & \ldots & u_{n-1} \end{pmatrix} \times \begin{pmatrix} p_n \\ p_{n-1} \\ \vdots \\ p_1 \end{pmatrix}$$

On pouvait donc se servir des fonctionnalités **Scilab** sur les matrices afin de répondre à cette question.

- Étant donné un vecteur ligne `U`, comment sélectionne-t-on les 5 premiers éléments de `U` ?

  `U(1,1:5)` ou tout simplement `U(1:5)`.

- Comment obtenir un vecteur colonne à partir d'un vecteur ligne `P`.

  On utilise l'opérateur de transposition. L'appel correspondant est `P'`.

- Comment peut-on obtenir le miroir d'un vecteur `V` comportant `n` éléments (*i.e.* un vecteur contenant les éléments de `V` rangés dans l'ordre inverse).

  Il suffit de réaliser l'appel `V(1,n:-1:1)` ou tout simplement `V(n:-1:1)`

- Écrire une nouvelle version de la fonction `calcSuiteU` en tirant profit des remarques précédentes.

```
1  function res = calcSuiteU(P)
2      [m, n] = size(P) // ou n = length(P)
3      U = zeros(1, n+1) // il y a n+1 entiers entre 0 et n
4      U(1) = 1 // la première case contient la valeur de u0
5      for i = 1:n
6          U(i+1) = U(1:i) * (P(i:-1:1))'
7      end
8      res = U(n+1)
9  endfunction
```

IV.3. ECRICOME - 2018

On considère les matrices :

$$A = \begin{pmatrix} 2 & 1 & -2 \\ 0 & 3 & 0 \\ 1 & -1 & 5 \end{pmatrix} \quad \text{et} \quad B = \begin{pmatrix} 1 & -1 & -1 \\ -3 & 3 & -3 \\ -1 & 1 & 1 \end{pmatrix}$$

On pose $X_0 = \begin{pmatrix} 3 \\ 0 \\ -1 \end{pmatrix}$, $X_1 = \begin{pmatrix} 3 \\ 0 \\ -2 \end{pmatrix}$, et pour tout entier naturel n : $X_{n+2} = \frac{1}{6}\, A\, X_{n+1} + \frac{1}{6}\, B\, X_n$.

a) Compléter la fonction ci-dessous qui prend en argument un entier n supérieur ou égal à 2 et qui renvoie la matrice X_n :

```
1  function res = X(n)
2      Xold = [3; 0; -1]
3      Xnew = [3; 0; -2]
4      A = [2,1,-2; 0,3,0; 1,-1,5]
5      B = [1,-1,-1; -3,3,-3; -1,1,1]
6      for i = 2:n
7          Aux = ..........
8          Xold = .........
9          Xnew = .........
10     end
11     res = .........
12 endfunction
```

Détaillons les différents éléments présents dans cette fonction.

- En ligne 2 et 3, on définit les variables `Xold` et `Xnew`.
  Ces deux variables sont initialement affectées aux valeurs X_0 et X_1.
- En ligne 4 et 5, on stocke les matrices A et B dans les variables `A` et `B`.
- De la ligne 6 à la ligne 10, on met à jour les variables `Xold` et `Xnew` de sorte à ce qu'elles contiennent les valeurs successives de la suite matricielle (X_n).

```
6      for i = 2:n
7          Aux = Xold
8          Xold = Xnew
9          Xnew = 1/6 * A * Xold + 1/6 * B * Aux
10     end
```

- Enfin, il n'y a plus qu'à affecter à `res` la valeur X_n.

```
11         res = Xnew
```

b) La fonction précédente a été utilisée dans un script permettant d'obtenir graphiquement (voir figure 1) les valeurs de α_n, β_n et γ_n en fonction de n.
Associer chacune des trois représentations graphiques à chacune des suites $(\alpha_n)_{n\in\mathbb{N}}$, $(\beta_n)_{n\in\mathbb{N}}$, $(\gamma_n)_{n\in\mathbb{N}}$ en justifiant votre réponse.



Soit $n \in \mathbb{N}$.

- D'après la question ***5*** :

$$\beta_n = \left(\frac{1}{2}\right)^{n-1} - \frac{2}{3}\left(-\frac{1}{2}\right)^{n} - \frac{4}{3} \underset{n\to+\infty}{\longrightarrow} -\frac{4}{3} \simeq -1,33$$

  En effet, comme $\left|\frac{1}{2}\right| < 1$ et $\left|-\frac{1}{2}\right| < 1$ alors : $\lim\limits_{n\to+\infty} \left(\frac{1}{2}\right)^{n-1} = 0$ et $\lim\limits_{n\to+\infty} \left(-\frac{1}{2}\right)^{n} = 0$.
- On démontre de même, toujours d'après les expressions obtenues en question ***5.*** :

$$\alpha_n \underset{n\to+\infty}{\longrightarrow} \frac{11}{6} \simeq 1,8 \quad \text{et} \quad \gamma_n \underset{n\to+\infty}{\longrightarrow} -\frac{11}{6} \simeq -1,8$$

- Or, d'après la figure :
  - × la suite repérée par × semble converger vers un réel valant approximativement $1,8$.

    Cette représentation graphique correspond à la suite (α_n).
  - × la suite repérée par $\oplus$ semble converger vers un réel valant approximativement $-1,3$.

    Cette représentation graphique correspond à la suite (β_n).
  - × la suite repérée par $\Diamond$ semble converger vers un réel valant approximativement $-1,8$.

    Cette représentation graphique correspond à la suite (γ_n).

IV.4. EML - 2018

On pose : $u_0 = 4$ et $\forall n \in \mathbb{N},\ u_{n+1} = \ln(u_n) + 2$.

▸ Écrire une fonction **Scilab** d'en-tête `function u = suite(n)` qui, prenant en argument un entier n de $\mathbb{N}$, renvoie la valeur de u_n.

```
1  function u = suite(n)
2      u = 4
3      for k = 1:n
4          u = log(u) + 2
5      end
6  endfunction
```

Expliquons un peu ce programme.
La variable `u` est créée pour contenir successivement les valeurs u_0, u_1, …, u_n.

- On initialise donc cette variable à $u_0 = 4$ avec la ligne 2 :

```
2      u = 4
```

- On met ensuite à jour `u` de manière itérative avec la ligne 4 :

```
4          u = log(u) + 2
```

Commentaire

Si on avait souhaité afficher tous les n premiers termes de la suite (u_n), on aurait modifié le script précédent de la façon suivante :

```
1  function u = suite(n)
2      u = zeros(1, n)
3      u(1) = 4
4      for k = 2:n
5          u(k) = log(u(k-1)) + 2
6      end
7  endfunction
```